

Birkhoff's Completeness Theorem for Multi-Sorted Algebras Formalized in Agda

Andreas Abel

Department of Computer Science, Gothenburg University, Sweden

28 May 2021

This document provides a formal proof of Birkhoff's completeness theorem for multi-sorted algebras which states that any equational entailment valid in all models is also provable in the equational theory. More precisely, if a certain equation is valid in all models that validate a fixed set of equations, then this equation is derivable from that set using the proof rules for a congruence.

The proof has been formalized in Agda version 2.6.2 with the Agda Standard Library version 1.7 and this document reproduces the commented Agda code.

1 Introduction

Birkhoff's completeness theorem [1935] has been formalized in type theory before, even in Agda [Gunther et al., 2017, Thm .3.1]. Our formalization makes the following decisions:

1. We use indexed containers [Altenkirch et al., 2015] aka Peterson-Synek (interaction) trees. Given a set S of sort symbols, a signature over S is an indexed endo-container, which has three components:
 - a) Per sort $s : S$, a set O_s of operator symbols. (In the container terminology, these are called *shapes* for index s , and in the interaction tree terminology, *commands* for state s .)
 - b) Per operator $o : O_s$, a set A_o , the arity of operator o . The arity is the index set for the arguments of the operator, which are then given by a function over A_o . (In the other terminologies, these are the *positions* or *responses*, resp.)
 - c) Per argument index $i : A_o$, a sort $s_i : S$ which denotes the sort of the i th argument of operator o . (In the interaction tree terminology, this is the *next* state.)

Closed terms of a multi-sorted algebra (aka first-order terms) are then concrete interaction trees, i.e., elements of the indexed W -type pertaining to the container.

Note that all the “set”s we mentioned above come with a size, see next point.

2. Universe-polymorphic: As we are working in a predicative and constructive meta-theory, we have to be aware of the size (i.e., inaccessible cardinality) of the sets. Our formalization is universe-polymorphic to ensure good generality, resting on the universe-polymorphic Agda Standard Library.

In particular, there is no such thing as “all models”; rather we can only quantify over models of a certain maximum size. The completeness theorem consequently does not require validity of an entailment in all models, but only in all models of a certain size, which is given by the size of the generic model, i.e., the term model. The size of the term model in turn is determined by the size of the signature of the multi-sorted algebra.

3. Open terms (with free variables) are obtained as the *free monad* over the container. Concretely, we make a new container that has additional nullary operator symbols, which stand for the variables. Terms are intrinsically typed, i.e., the set of terms is actually a family of sets indexed by a sort and a context of sorted variables in scope.
4. No lists: We have no finiteness restrictions whatsoever, neither the number of operators need to be finite, nor the number of arguments of an operator, nor the set of variables that are in scope of a term. (Note however, since terms are finite trees, they can actually mention only a finite number of variables from the possibly infinite supply.)

2 Preliminaries

We import library content for indexed containers, standard types, and setoids.

```
open import Level

open import Data.Container.Indexed          using (Container; [ ]; μ; _◁_/_)
open import Data.Container.Indexed.FreeMonad using (_★C_)
open import Data.W.Indexed                  using (sup)

open import Data.Product                    using (Σ; _×_; _,_; Σ-syntax); open Σ
open import Data.Sum                        using (_⊔_; inj1; inj2; [_, _])
open import Data.Empty.Polymorphic          using (⊥; ⊥-elim)

open import Function                        using (_◦_)
open import Function.Bundles                using (Func)

open import Relation.Binary                 using (Setoid; IsEquivalence)
open import Relation.Binary.PropositionalEquality using (_≡_; refl)

import Relation.Binary.Reasoning.Setoid as SetoidReasoning

open Setoid using (Carrier; _≈_; isEquivalence)
open Func renaming (f to apply)
```

Letter ℓ denotes universe levels.

variable

$\ell \ell' \ell^s \ell^o \ell^a \ell^m \ell^e \ell^i : \text{Level}$

$I : \text{Set } \ell^i$

$S : \text{Set } \ell^s$

The interpretation of a container ($\text{Op} \triangleleft \text{Ar} / \text{sort}$) is

$\llbracket \text{Op} \triangleleft \text{Ar} / \text{sort} \rrbracket X s = \Sigma [o \in \text{Op } s] ((i : \text{Ar } o) \rightarrow X (\text{sort } o \ i))$

which contains pairs consisting of an operator o and its collection of arguments. The least fixed point of $(X \mapsto \llbracket C \rrbracket X)$ is the indexed W-type given by C , and it contains closed first-order terms of the multi-sorted algebra C .

We need to interpreting indexed containers on Setoids. This definition is missing from the standard library v1.7. It equips the sets $(\llbracket C \rrbracket X s)$ with an equivalence relation induced by the one of the family X . The definition of $\llbracket _ \rrbracket s$ can be stated for heterogeneous index containers where we distinguish input and output sorts I and S .

$\llbracket _ \rrbracket s : (C : \text{Container } I S \ell^o \ell^a) (\xi : I \rightarrow \text{Setoid } \ell^m \ell^e) \rightarrow S \rightarrow \text{Setoid } _ _$

$\llbracket C \rrbracket s \xi s . \text{Carrier} =$
 $\llbracket C \rrbracket (\text{Carrier} \circ \xi) s$

$\llbracket \text{Op} \triangleleft \text{Ar} / \text{sort} \rrbracket s \xi s . _ \approx _ (op, args) (op', args') =$
 $\Sigma [eq \in op \equiv op'] \text{EqArgs } eq \ args \ args'$

where

$\text{EqArgs} : (eq : op \equiv op')$

$(args : (i : \text{Ar } op) \rightarrow \xi (\text{sort } _ \ i) . \text{Carrier})$

$(args' : (i : \text{Ar } op') \rightarrow \xi (\text{sort } _ \ i) . \text{Carrier})$

$\rightarrow \text{Set } _$

$\text{EqArgs refl } args \ args' = (i : \text{Ar } op) \rightarrow \xi (\text{sort } _ \ i) . _ \approx _ (args \ i) (args' \ i)$

$\llbracket \text{Op} \triangleleft \text{Ar} / \text{sort} \rrbracket s \xi s . \text{isEquivalence} . \text{isEquivalence.refl}$
 $= \text{refl} , \lambda i \rightarrow \text{Setoid.refl } (\xi (\text{sort } _ \ i))$

$\llbracket \text{Op} \triangleleft \text{Ar} / \text{sort} \rrbracket s \xi s . \text{isEquivalence} . \text{isEquivalence.sym } (\text{refl} , g)$
 $= \text{refl} , \lambda i \rightarrow \text{Setoid.sym } (\xi (\text{sort } _ \ i)) (g \ i)$

$\llbracket \text{Op} \triangleleft \text{Ar} / \text{sort} \rrbracket s \xi s . \text{isEquivalence} . \text{isEquivalence.trans } (\text{refl} , g) (\text{refl} , h)$
 $= \text{refl} , \lambda i \rightarrow \text{Setoid.trans } (\xi (\text{sort } _ \ i)) (g \ i) (h \ i)$

3 Multi-sorted algebras

A multi-sorted algebra is an indexed container.

- Sorts are indexes.
- Operators are commands/shapes.

- Arities/argument are responses/positions.

Closed terms (initial model) are given by the W type for a container, renamed to μ here (for least fixed-point).

We assume a fixed signature (Sort, Ops).

```
module _ (Sort : Set  $\ell^s$ ) (Ops : Container Sort Sort  $\ell^o$   $\ell^a$ ) where
  open Container Ops renaming
    ( Command to Op
      ; Response to Arity
      ; next to sort
    )
```

We let letter s range over sorts and op over operators.

```
variable
  s s' : Sort
  op op' : Op s
```

3.1 Models

A model is given by an interpretation (Den s) for each sort s plus an interpretation (den o) for each operator o .

```
record SetModel  $\ell^m$  : Set ( $\ell^s \sqcup \ell^o \sqcup \ell^a \sqcup \text{succ } \ell^m$ ) where
  field
    Den : Sort  $\rightarrow$  Set  $\ell^m$ 
    den : {s : Sort}  $\rightarrow$  [Ops] Den s  $\rightarrow$  Den s
```

The setoid model requires operators to respect equality. The Func record packs a function (apply) with a proof (cong) that the function maps equals to equals.

```
record SetoidModel  $\ell^m \ell^e$  : Set ( $\ell^s \sqcup \ell^o \sqcup \ell^a \sqcup \text{succ } (\ell^m \sqcup \ell^e)$ ) where
  field
    Den : Sort  $\rightarrow$  Setoid  $\ell^m \ell^e$ 
    den : {s : Sort}  $\rightarrow$  Func ([Ops] s Den s) (Den s)
```

4 Terms

To obtain terms with free variables, we add additional nullary operators, each representing a variable.

These are covered in the standard library `FreeMonad` module, albeit with the restriction that the operator and variable sets have the same size.

$$\text{Cxt} = \text{Sort} \rightarrow \text{Set } \ell^0$$

$$\begin{array}{l} \text{variable} \\ \Gamma \Delta : \text{Cxt} \end{array}$$

Terms with free variables in `Var`.

$$\text{module } _ (\text{Var} : \text{Cxt}) \text{ where}$$

We keep the same sorts, but add a nullary operator for each variable.

$$\begin{array}{l} \text{Ops}^+ : \text{Container } \text{Sort } \text{Sort } \ell^0 \ell^a \\ \text{Ops}^+ = \text{Ops} \star \text{C } \text{Var} \end{array}$$

Terms with variables are then given by the W -type (named μ) for the extended container.

$$\text{Tm} = \mu \text{Ops}^+$$

We define nice constructors for variables and operator application via pattern synonyms. Note that the f in constructor `var'` is a function from the empty set, so it should be uniquely determined. However, Agda's equality is more intensional and will not identify all functions from the empty set. Since we do not make use of the axiom of function extensionality, we sometimes have to consult the extensional equality of the function setoid.

$$\begin{array}{ll} \text{pattern } _ \bullet _ \text{ op } \text{ args} & = \text{sup } (\text{inj}_2 \text{ op } , \text{ args}) \\ \text{pattern } \text{var}' \ x \ f & = \text{sup } (\text{inj}_1 \ x , f) \\ \text{pattern } \text{var} \ x & = \text{var}' \ x \ _ \end{array}$$

Letter t ranges over terms, and ts over argument vectors.

$$\begin{array}{l} \text{variable} \\ t \ t' \ t_1 \ t_2 \ t_3 : \text{Tm } \Gamma \ s \\ ts \ ts' : (i : \text{Arity } \text{op}) \rightarrow \text{Tm } \Gamma (\text{sort } _ i) \end{array}$$

4.1 Parallel substitutions

A substitution from Δ to Γ holds a term in Γ for each variable in Δ .

$$\begin{array}{l} \text{Sub} : (\Gamma \Delta : \text{Cxt}) \rightarrow \text{Set } _ \\ \text{Sub } \Gamma \Delta = \forall \{s\} (x : \Delta \ s) \rightarrow \text{Tm } \Gamma \ s \end{array}$$

Application of a substitution.

$$\begin{aligned} \llbracket _ \rrbracket &: (t : \mathbf{Tm} \Delta s) (\sigma : \mathbf{Sub} \Gamma \Delta) \rightarrow \mathbf{Tm} \Gamma s \\ (\mathbf{var} \ x) \llbracket \sigma \rrbracket &= \sigma \ x \\ (op \bullet ts) \llbracket \sigma \rrbracket &= op \bullet \lambda i \rightarrow ts \ i \llbracket \sigma \rrbracket \end{aligned}$$

Letter σ ranges over substitutions.

variable
 $\sigma \ \sigma' : \mathbf{Sub} \Gamma \Delta$

5 Interpretation of terms in a model

Given an algebra M of set-size ℓ^m and equality-size ℓ^e , we define the interpretation of an s -sorted term t as element of $M(s)$ according to an environment ρ that maps each variable of sort s' to an element of $M(s')$.

module $_ \{M : \mathbf{SetoidModel} \ell^m \ell^e\}$ where
 open $\mathbf{SetoidModel} \ M$

Equality in M 's interpretation of sort s .

$$\begin{aligned} _ \approx _ &: \mathbf{Den} \ s \ . \mathbf{Carrier} \rightarrow \mathbf{Den} \ s \ . \mathbf{Carrier} \rightarrow \mathbf{Set} \ _ \\ _ \approx _ \{s = s\} &= \mathbf{Den} \ s \ . _ \approx _ \end{aligned}$$

An environment for Γ maps each variable $x : \Gamma(s)$ to an element of $M(s)$. Equality of environments is defined pointwise.

$$\begin{aligned} \mathbf{Env} &: \mathbf{Cxt} \rightarrow \mathbf{Setoid} \ _ _ \\ \mathbf{Env} \ \Gamma \ . \mathbf{Carrier} &= \{s : \mathbf{Sort}\} (x : \Gamma \ s) \rightarrow \mathbf{Den} \ s \ . \mathbf{Carrier} \\ \mathbf{Env} \ \Gamma \ . _ \approx _ \rho \ \rho' &= \{s : \mathbf{Sort}\} (x : \Gamma \ s) \rightarrow \rho \ x \approx \rho' \ x \\ \mathbf{Env} \ \Gamma \ . \mathbf{isEquivalence} \ . \mathbf{IsEquivalence.refl} &\{s = s\} \ x = \mathbf{Den} \ s \ . \mathbf{Setoid.refl} \\ \mathbf{Env} \ \Gamma \ . \mathbf{isEquivalence} \ . \mathbf{IsEquivalence.sym} &\ h \ \{s\} \ x = \mathbf{Den} \ s \ . \mathbf{Setoid.sym} \ (h \ x) \\ \mathbf{Env} \ \Gamma \ . \mathbf{isEquivalence} \ . \mathbf{IsEquivalence.trans} &\ g \ h \ \{s\} \ x = \mathbf{Den} \ s \ . \mathbf{Setoid.trans} \ (g \ x) \ (h \ x) \end{aligned}$$

Interpretation of terms is iteration on the W-type. The standard library offers ‘iter’ (on sets), but we need this to be a Func (on setoids).

$$\begin{aligned} (_) &: \forall \{s\} (t : \mathbf{Tm} \Gamma s) \rightarrow \mathbf{Func} (\mathbf{Env} \ \Gamma) (\mathbf{Den} \ s) \\ (\mathbf{var} \ x \) \ . \mathbf{apply} \ \rho &= \rho \ x \\ (\mathbf{var} \ x \) \ . \mathbf{cong} \ \rho = \rho' &= \rho = \rho' \ x \\ (op \bullet args) \ . \mathbf{apply} \ \rho &= \mathbf{den} \ . \mathbf{apply} \ (op \ , \ \lambda i \rightarrow (\mathbf{args} \ i) \ . \mathbf{apply} \ \rho) \\ (op \bullet args) \ . \mathbf{cong} \ \rho = \rho' &= \mathbf{den} \ . \mathbf{cong} \ (\mathbf{refl} \ , \ \lambda i \rightarrow (\mathbf{args} \ i) \ . \mathbf{cong} \ \rho = \rho') \end{aligned}$$

An equality between two terms holds in a model if the two terms are equal under all valuations of their free variables.

```
Equal : ∀ {Γ s} (t t' : Tm Γ s) → Set _
Equal {Γ} {s} t t' = ∀ (ρ : Env Γ .Carrier) → ((t) .apply ρ) ≈ ((t') .apply ρ)
```

This notion is an equivalence relation.

```
isEquiv : IsEquivalence (Equal {Γ = Γ} {s = s})
isEquiv {s = s} .IsEquivalence.refl ρ = Den s .Setoid.refl
isEquiv {s = s} .IsEquivalence.sym e ρ = Den s .Setoid.sym (e ρ)
isEquiv {s = s} .IsEquivalence.trans e e' ρ = Den s .Setoid.trans (e ρ) (e' ρ)
```

5.1 Substitution lemma

Evaluation of a substitution gives an environment.

```
((_)s : Sub Γ Δ → Env Γ .Carrier → Env Δ .Carrier
(σ)s ρ x = ((σ x) .apply ρ)
```

Substitution lemma: $((t[\sigma]))\rho \approx ((t))((\sigma)\rho)$

```
substitution : (t : Tm Δ s) (σ : Sub Γ Δ) (ρ : Env Γ .Carrier) →
  ((t [ σ ])) .apply ρ ≈ ((t) .apply ((σ)s ρ))
substitution (var x) σ ρ = Den _ .Setoid.refl
substitution (op • ts) σ ρ = den .cong (refl, λ i → substitution (ts i) σ ρ)
```

6 Equations

An equation is a pair $t \doteq t'$ of terms of the same sort in the same context.

```
record Eq : Set (ℓs ⊔ suc ℓo ⊔ ℓa) where
  constructor _≐_
  field
    {cxt} : Sort → Set ℓo
    {srt} : Sort
    lhs   : Tm cxt srt
    rhs   : Tm cxt srt
```

Equation $t \doteq t'$ holding in model M .

```
_F_ : (M : SetoidModel ℓm ℓe) (eq : Eq) → Set _
M ⊨ (t ≐ t') = Equal {M = M} t t'
```

Sets of equations are presented as collection $E : I \rightarrow \text{Eq}$ for some index set $I : \text{Set } \ell^i$.

An entailment/consequence $E \supset t \doteq t'$ is valid if $t \doteq t'$ holds in all models that satisfy equations E .

module $_ \{ \ell^m \ell^e \}$ **where**

$_ \supset _ : (E : I \rightarrow \text{Eq}) (eq : \text{Eq}) \rightarrow \text{Set } _$
 $E \supset eq = \forall (M : \text{SetoidModel } \ell^m \ell^e) \rightarrow (\forall i \rightarrow M \models E i) \rightarrow M \models eq$

6.1 Derivations

Equational logic allows us to prove entailments via the inference rules for the judgment $E \vdash \Gamma \triangleright t \equiv t'$. This could be coined as equational theory over a given set of equations E . Relation $E \vdash \Gamma \triangleright _ \equiv _$ is the least congruence over the equations E .

data $_ \vdash _ \triangleright _ \equiv _ \{ I : \text{Set } \ell^i \}$
 $(E : I \rightarrow \text{Eq}) : (\Gamma : \text{Cxt}) (t t' : \text{Tm } \Gamma s) \rightarrow \text{Set } (\ell^s \sqcup \text{suc } \ell^o \sqcup \ell^a \sqcup \ell^i)$ **where**

hyp : $\forall i \rightarrow \text{let } t \doteq t' = E i \text{ in}$
 $E \vdash _ \triangleright t \equiv t'$

base : $\forall (x : \Gamma s) \{ ff' : (i : \perp) \rightarrow \text{Tm } _ (\perp\text{-elim } i) \} \rightarrow$
 $E \vdash \Gamma \triangleright \text{var}' x f \equiv \text{var}' x f'$

app : $(\forall i \rightarrow E \vdash \Gamma \triangleright ts i \equiv ts' i) \rightarrow$
 $E \vdash \Gamma \triangleright (op \bullet ts) \equiv (op \bullet ts')$

sub : $E \vdash \Delta \triangleright t \equiv t' \rightarrow$
 $\forall (\sigma : \text{Sub } \Gamma \Delta) \rightarrow$
 $E \vdash \Gamma \triangleright (t [\sigma]) \equiv (t' [\sigma])$

refl : $\forall (t : \text{Tm } \Gamma s) \rightarrow$
 $E \vdash \Gamma \triangleright t \equiv t$

sym : $E \vdash \Gamma \triangleright t \equiv t' \rightarrow$
 $E \vdash \Gamma \triangleright t' \equiv t$

trans : $E \vdash \Gamma \triangleright t_1 \equiv t_2 \rightarrow$
 $E \vdash \Gamma \triangleright t_2 \equiv t_3 \rightarrow$
 $E \vdash \Gamma \triangleright t_1 \equiv t_3$

6.2 Soundness of the inference rules

We assume a model M that validates all equations in E .

```
module Soundness {I : Set ℓi} (E : I → Eq) (M : SetoidModel ℓm ℓe)
  (V : ∀ i → M ⊨ E i) where
  open SetoidModel M
```

In any model M that satisfies the equations E , derived equality is actual equality.

```
sound : E ⊢ Γ ▷ t ≡ t' → M ⊨ (t ≐ t')

sound (hyp i) = V i
sound (app {op = op} es) ρ = den .cong (refl, λ i → sound (es i) ρ)
sound (sub {t = t} {t' = t'} e σ) ρ = begin
  ((t [σ] ) .apply ρ ≈⟨ substitution {M = M} t σ ρ ⟩
   (t      ) .apply ρ' ≈⟨ sound e ρ' ⟩
   (t'     ) .apply ρ' ≈~⟨ substitution {M = M} t' σ ρ ⟩
   (t' [σ] ) .apply ρ) ■
  where
  open SetoidReasoning (Den _)
  ρ' = (σ) s ρ

sound (base x {f} {f'}) = isEquiv {M = M} .IsEquivalence.refl {var' x λ()}

sound (refl t) = isEquiv {M = M} .IsEquivalence.refl {t}
sound (sym {t = t} {t' = t'} e) = isEquiv {M = M} .IsEquivalence.sym
                                   {x = t} {y = t'} (sound e)

sound (trans {t1 = t1} {t2 = t2}
             {t3 = t3} e e') = isEquiv {M = M} .IsEquivalence.trans
                                   {i = t1} {j = t2} {k = t3} (sound e) (sound e')
```

7 Birkhoff's completeness theorem

Birkhoff proved that any equation $t \doteq t'$ is derivable from E when it is valid in all models satisfying E . His proof (for single-sorted algebras) is a blue print for many more completeness proofs. They all proceed by constructing a universal model aka term model. In our case, it is terms quotiented by derivable equality $E \vdash \Gamma \triangleright _ \equiv _$. It then suffices to prove that this model satisfies all equations in E .

7.1 Universal model

A term model for E and Γ interprets sort s by $(\text{Tm } \Gamma \ s)$ quotiented by $E \vdash \Gamma \triangleright _ \equiv _$.

```

module TermModel {I : Set ℓi} (E : I → Eq) where
  open SetoidModel

```

Tm Γ s quotiented by $E \vdash \Gamma \triangleright \cdot \equiv \cdot$.

```

TmSetoid : Cxt → Sort → Setoid _ _
TmSetoid  $\Gamma$  s .Carrier = Tm  $\Gamma$  s
TmSetoid  $\Gamma$  s . $\approx$  =  $E \vdash \Gamma \triangleright \cdot \equiv \cdot$ 
TmSetoid  $\Gamma$  s .isEquivalence .isEquivalence.refl = refl _
TmSetoid  $\Gamma$  s .isEquivalence .isEquivalence.sym = sym
TmSetoid  $\Gamma$  s .isEquivalence .isEquivalence.trans = trans

```

The interpretation of an operator is simply the operator. This works because $E \vdash \Gamma \triangleright \cdot \equiv \cdot$ is a congruence.

```

tmlInterp : ∀ { $\Gamma$  s} → Func (⟦ Ops ⟧s (TmSetoid  $\Gamma$  s) (TmSetoid  $\Gamma$  s))
tmlInterp .apply (op , ts) = op • ts
tmlInterp .cong (refl , h) = app h

```

The term model per context Γ .

```

M : Cxt → SetoidModel _ _
M  $\Gamma$  .Den = TmSetoid  $\Gamma$ 
M  $\Gamma$  .den = tmlInterp

```

The identity substitution σ_0 maps variables to themselves.

```

 $\sigma_0$  : { $\Gamma$  : Cxt} → Sub  $\Gamma$   $\Gamma$ 
 $\sigma_0$  x = var' x  $\lambda$ ()

```

σ_0 acts indeed as identity.

```

identity : (t : Tm  $\Gamma$  s) →  $E \vdash \Gamma \triangleright t [\sigma_0] \equiv t$ 
identity (var x) = base x
identity (op • ts) = app  $\lambda$  i → identity (ts i)

```

Evaluation in the term model is substitution $E \vdash \Gamma \triangleright (t)\sigma \equiv t[\sigma]$. This would even hold "up to the nose" if we had function extensionality.

```

evaluation : (t : Tm  $\Delta$  s) (σ : Sub  $\Gamma$   $\Delta$ ) →  $E \vdash \Gamma \triangleright ((\_) \{M = M \Gamma\} t .apply \sigma) \equiv (t [\sigma])$ 
evaluation (var x) σ = refl (σ x)
evaluation (op • ts) σ = app ( $\lambda$  i → evaluation (ts i) σ)

```

The term model satisfies all the equations it started out with.

```

satisfies :  $\forall i \rightarrow M \Gamma \vdash E i$ 
satisfies  $i \sigma =$  begin
  ( $t_l$ ) .apply  $\sigma \approx \langle \text{evaluation } t_l \sigma \rangle$ 
   $t_l [\sigma] \approx \langle \text{sub (hyp } i) \sigma \rangle$ 
   $t_r [\sigma] \approx \langle \text{evaluation } t_r \sigma \rangle$ 
  ( $t_r$ ) .apply  $\sigma$  ■
where
open SetoidReasoning (TmSetoid  $\_ \_$ )
 $t_l = E i .\text{Eq.lhs}$ 
 $t_r = E i .\text{Eq.rhs}$ 

```

7.2 Completeness

Birkhoff's completeness theorem [1935]: Any valid consequence is derivable in the equational theory.

```

module Completeness { $I : \text{Set } \mathcal{C}^i$ } ( $E : I \rightarrow \text{Eq}$ ) { $\Gamma s$ } { $t t' : \text{Tm } \Gamma s$ } where
open TermModel E

completeness :  $E \supset (t \doteq t') \rightarrow E \vdash \Gamma \triangleright t \equiv t'$ 
completeness  $V =$  begin
   $t \approx \langle \text{identity } t \rangle$ 
   $t [\sigma_0] \approx \langle \text{evaluation } t \sigma_0 \rangle$ 
  ( $t$ ) .apply  $\sigma_0 \approx \langle V (M \Gamma) \text{ satisfies } \sigma_0 \rangle$ 
  ( $t'$ ) .apply  $\sigma_0 \approx \langle \text{evaluation } t' \sigma_0 \rangle$ 
   $t' [\sigma_0] \approx \langle \text{identity } t' \rangle$ 
   $t'$  ■
where open SetoidReasoning (TmSetoid  $\Gamma s$ )

```

Q.E.D.

8 Related work

Gunther et al. [2017] further formalize signature morphisms. These would be, in our setting, morphisms of indexed containers, described by Altenkirch et al. [2015], albeit in a slightly different semantics, slice categories.

DeMeo's rather comprehensive development [2021] formalizes single-sorted algebras up the Birkhoff's HSP theorem in Agda. DeMeo's signatures are containers, even though he does not make this connection explicit. DeMeo's formalization is basis for <https://github.com/ualib/agda-algebras>.

Acknowledgments. This document has been generated from `MultiSortedAlgebra.agda` using the `agda2lagda` translator and the `agda --latex` backend.

References

- T. Altenkirch, N. Ghani, P. G. Hancock, C. McBride, and P. Morris. Indexed containers. *J. Func. Program.*, 25, 2015. doi: 10.1017/S095679681500009X. URL <https://doi.org/10.1017/S095679681500009X>.
- G. Birkhoff. On the structure of abstract algebras. *Mathematical Proceedings of the Cambridge Philosophical Society*, 31(4):433–454, 1935. doi: 10.1017/S0305004100013463.
- W. J. DeMeo. The Agda Universal Algebra Library and Birkhoff’s theorem in Martin-Löf dependent type theory. *CoRR*, abs/2101.10166, 2021. URL <https://arxiv.org/abs/2101.10166>.
- E. Gunther, A. Gadea, and M. Pagano. Formalization of universal algebra in Agda. In S. Alves and R. Wasserman, editors, *12th Workshop on Logical and Semantic Frameworks, with Applications, LSFA 2017, Brasília, Brazil, September 23-24, 2017*, volume 338 of *Electr. Notes in Theor. Comp. Sci.*, pages 147–166. Elsevier, 2017. doi: 10.1016/j.entcs.2018.10.010. URL <https://doi.org/10.1016/j.entcs.2018.10.010>.